

Using Style Elements

You can use style elements to control attributes of plot features such as markers and lines as follows:

```
proc sgplot data=sashelp.class noautolegend;
  pbspline y=weight x=height /
  markerattrs=GraphData2
  lineattrs=GraphFit2;
run;
```

You can set most attributes by using style elements and override certain attributes with options as follows:

```
proc sgplot data=sashelp.class noautolegend;
  pbspline y=weight x=height /
  markerattrs=Graphdata2(symbol=circlefilled)
  lineattrs=GraphFit2(thickness=1px);
run;
```

In the GTL, you can specify both style elements and particular attributes as follows:

```
proc template;
  define statgraph pbs;
    begingraph;
    layout overlay;
    scatterplot y=weight x=height /
    markerattrs=GraphData5
    (color=GraphData2:ContrastColor);
    pbsplineplot y=weight x=height /
    lineattrs=GraphData8
    (color=GraphFit2:Color);
    endlayout;
  endgraph;
end;
run;
```

```
proc sgrender data=sashelp.class template=pbs;
run;
```

The marker style comes from GraphData5 but with the GraphData2 contrast color. The line style comes from GraphData8 but with the GraphFit2 color.


Style Template Creation Example

```
proc template;
  define Style styles.MyStyle;
    parent=styles.htmlblue;

    class GraphDataDefault /
      Color = cxCAD5E5
      ContrastColor = cx003399
      LineStyle = 1
      LineThickness = 1px
      MarkerSymbol = "StarFilled"
      MarkerSize = 5px
      StartColor = cxAFB5A6
      NeutralColor = cxFFFFFF
      EndColor = cx667FA2;

    style colors from colors /
      'docbg' = cxFABFE;
  end;
run;

ods html style=mystyle;
proc sgplot data=sashelp.class noautolegend;
  title 'Polynomial Fit';
  reg y=weight x=height / degree=3;
run;
ods html close;
```



For more information, see the following:

Basic ODS Graphics Examples
<http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsbasicq.pdf>

Advanced ODS Graphics Examples
<http://support.sas.com/documentation/prod-p/grstat/9.4/en/PDF/odsadvq.pdf>

For complete information, see the SAS 9.4 documentation at
<http://support.sas.com/documentation/>.

Graph Style Tip Sheet

This tip sheet collects frequently used information in one place so you don't have to search through the online documentation. It also gives you examples to take home and try.

ODS Graphics is an extension of ODS (the Output Delivery System). ODS styles control the colors and general appearance of all graphs and tables, and SAS[®] provides several styles that are recommended for use with statistical graphics. The default style that you see when you run SAS depends on the ODS destination, system options, and SAS registry settings.

This tip sheet presents an overview of ODS styles and style elements and their use in ODS Graphics in SAS 9.4.


SAS⁹ Graph Style Tip Sheet

ODS Styles

Style (Default in Destination or Application)


HTMLBlue (HTML)

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001




Pearl (PDF)

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001




RTF (RTF)

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001




PearlJ (SAS/STAT[®] documentation)

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001




Analysis (SAS[®] Enterprise Guide[®])

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001



Journal

Parameter Estimates				
Variable	DF	Parameter Estimate	Standard Error	t Value Pr > t
Intercept	1	-143.02692	32.27459	-4.43 0.0004
Height	1	3.89903	0.51609	7.55 <.0001



Displaying Styles

Use PROC TEMPLATE to display style definitions:

```
proc template;
  list styles;
  * list all styles;
  * list a style and its parents;
  source styles.htmlblue / expand;*style & parents;
run;
```

Line Styles, Markers, Colors

LineStyle numbers: Dash 20, DashDashDot 14, DashDotDot 15, Dot 34, LongDash 5, LongDashShortDash 26, MediumDash 4, MediumDashDotDot 42, MediumDashShortDash 8, ShortDash 2, ShortDashDot 41, Solid 1, ThinDot 35


MarkerSymbol values: ArrowDown Asterisk Circle CircleFilled Diamond DiamondFilled GreaterThan Hash HomeDown HomeDownFilled IBeam Plus Square SquareFilled Star StarFilled Tack Tilde Triangle TriangleFilled Union X Y Z

Colors: Color is for filled areas, and contrast color is for markers and lines. Colors can be specified as *CXrrggb*, where the last six characters specify RGB (red, green, blue) values on the hexadecimal scale of 00 to FF (0 to 255 base 10).

Style Override

You can override the group style elements:

```
ods graphics on / attrpriority=none;
proc sgplot data=sashelp.iris;
  styleattrs datacontrastcolors=(magenta orange brown)
  datasymbols=(star square triangle);
  scatter y=petalwidth x=petallength /
  group=species;
run;
```



Graph Style Elements

You refer to ODS style elements in graph templates with *style-element* or *style-element.attribute* (for example, GraphDataDefault:ContrastColor). Common graph style elements follow:

Graph graph size, outer border appearance
GraphConfidence primary fit confidence interval
GraphData1 first grouped data item attributes
 . . .
GraphData12 twelfth grouped data item attributes
GraphDataDefault ungrouped data item attributes
GraphFit primary fit function
GraphFit2 secondary fit function
GraphGridLines horizontal and vertical grid lines
GraphOutlier outlier data attributes
GraphPredictionLimits fills for prediction limits
GraphReference reference lines and drop lines
GraphDataText attributes of point and line labels
GraphValueText attributes of axis tick values
GraphLabelText attr of axis labels and legend title
GraphFootnoteText footnotes
GraphTitleText titles
GraphWalls vertical walls bounded by axes

Common attributes and sample values follow:

```
BackgroundColor = colors("docbg")
Color = GraphColors("gdata")
ContrastColor = GraphColors("gcdata")
Displayopts = "fill outline"
EndColor = GraphColors("gconramp3cend")
Font = Fonts("TitleFont")
FrameBorder = on
LineStyle = 1
LineThickness = 3px
MarkerSize = 7px
MarkerSymbol = "circle"
NeutralColor = GraphColors("gconramp3neutral")
Padding = 7
StartColor = GraphColors("gconramp3cstart")
```

Many attributes are defined indirectly by default. Not all attributes can be used with all elements.

Graph Fonts

Some of the HTMLBlue style graph fonts follow:

```
style GraphFonts /
  'GraphDataFont' = ("

```

You can use PROC TEMPLATE with the SOURCE statement to display other styles and view other font definitions. You can create a new style by using modified font definitions as follows:

```
proc template;
  define style myfonts;
    parent = styles.htmlblue;
    style GraphFonts from GraphFonts /
      'GraphDataFont' = ("Arial",7pt)
      'GraphLabelFont' = ("Times New Roman",
        10pt)
      'GraphFootnoteFont' = ("Courier New",
        10pt);
  end;
run;
```

The style MYSTYLE differs from the HTMLBlue style only in the three font definitions. Specify STYLE=MYSTYLE in an ODS destination statement to use your new style.